To our customers,

Old Company Name in Catalogs and Other Documents

On April 1st, 2010, NEC Electronics Corporation merged with Renesas Technology Corporation, and Renesas Electronics Corporation took over all the business of both companies. Therefore, although the old company name remains in this document, it is a valid Renesas Electronics document. We appreciate your understanding.

Renesas Electronics website: http://www.renesas.com

April 1st, 2010
Renesas Electronics Corporation

Issued by: Renesas Electronics Corporation (http://www.renesas.com)

Send any inquiries to http://www.renesas.com/inquiry.

Notice

1. All information included in this document is current as of the date this document is issued. Such information, however, is subject to change without any prior notice. Before purchasing or using any Renesas Electronics products listed herein, please confirm the latest product information with a Renesas Electronics sales office. Also, please pay regular and careful attention to additional and different information to be disclosed by Renesas Electronics such as that disclosed through our website.
2. Renesas Electronics does not assume any liability for infringement of patents, copyrights, or other intellectual property rights of third parties by or arising from the use of Renesas Electronics products or technical information described in this document. No license, express, implied or otherwise, is granted hereby under any patents, copyrights or other intellectual property rights of Renesas Electronics or others.
3. You should not alter, modify, copy, or otherwise misappropriate any Renesas Electronics product, whether in whole or in part.
4. Descriptions of circuits, software and other related information in this document are provided only to illustrate the operation of semiconductor products and application examples. You are fully responsible for the incorporation of these circuits, software, and information in the design of your equipment. Renesas Electronics assumes no responsibility for any losses incurred by you or third parties arising from the use of these circuits, software, or information.
5. When exporting the products or technology described in this document, you should comply with the applicable export control laws and regulations and follow the procedures required by such laws and regulations. You should not use Renesas Electronics products or the technology described in this document for any purpose relating to military applications or use by the military, including but not limited to the development of weapons of mass destruction. Renesas Electronics products and technology may not be used for or incorporated into any products or systems whose manufacture, use, or sale is prohibited under any applicable domestic or foreign laws or regulations.
6. Renesas Electronics has used reasonable care in preparing the information included in this document, but Renesas Electronics does not warrant that such information is error free. Renesas Electronics assumes no liability whatsoever for any damages incurred by you resulting from errors in or omissions from the information included herein.
7. Renesas Electronics products are classified according to the following three quality grades: “Standard”, “High Quality”, and “Specific”. The recommended applications for each Renesas Electronics product depends on the product’s quality grade, as indicated below. You must check the quality grade of each Renesas Electronics product before using it in a particular application. You may not use any Renesas Electronics product for any application categorized as “Specific” without the prior written consent of Renesas Electronics. Further, you may not use any Renesas Electronics product for any application for which it is not intended without the prior written consent of Renesas Electronics. Renesas Electronics shall not be in any way liable for any damages or losses incurred by you or third parties arising from the use of any Renesas Electronics product for an application categorized as “Specific” or for which the product is not intended where you have failed to obtain the prior written consent of Renesas Electronics. The quality grade of each Renesas Electronics product is “Standard” unless otherwise expressly specified in a Renesas Electronics data sheets or data books, etc.

   “Standard”: Computers; office equipment; communications equipment; test and measurement equipment; audio and visual equipment; home electronic appliances; machine tools; personal electronic equipment; and industrial robots.

   “High Quality”: Transportation equipment (automobiles, trains, ships, etc.); traffic control systems; anti-disaster systems; anti-crime systems; safety equipment; and medical equipment not specifically designed for life support.

   “Specific”: Aircraft; aerospace equipment; submersible repeaters; nuclear reactor control systems; medical equipment or systems for life support (e.g. artificial life support devices or systems), surgical implantations, or healthcare intervention (e.g. excision, etc.), and any other applications or purposes that pose a direct threat to human life.
8. You should use the Renesas Electronics products described in this document within the range specified by Renesas Electronics, especially with respect to the maximum rating, operating supply voltage range, movement power voltage range, heat radiation characteristics, installation and other product characteristics. Renesas Electronics shall have no liability for malfunctions or damages arising out of the use of Renesas Electronics products beyond such specified ranges.
9. Although Renesas Electronics endeavors to improve the quality and reliability of its products, semiconductor products have specific characteristics such as the occurrence of failure at a certain rate and malfunctions under certain use conditions. Further, Renesas Electronics products are not subject to radiation resistance design. Please be sure to implement safety measures to guard them against the possibility of physical injury, and injury or damage caused by fire in the event of the failure of a Renesas Electronics product, such as safety design for hardware and software including but not limited to redundancy, fire control and malfunction prevention, appropriate treatment for aging degradation or any other appropriate measures. Because the evaluation of microcomputer software alone is very difficult, please evaluate the safety of the final products or system manufactured by you.
10. Please contact a Renesas Electronics sales office for details as to environmental matters such as the environmental compatibility of each Renesas Electronics product. Please use Renesas Electronics products in compliance with all applicable laws and regulations that regulate the inclusion or use of controlled substances, including without limitation, the EU RoHS Directive. Renesas Electronics assumes no liability for damages or losses occurring as a result of your noncompliance with applicable laws and regulations.
11. This document may not be reproduced or duplicated, in any form, in whole or in part, without prior written consent of Renesas Electronics.
12. Please contact a Renesas Electronics sales office if you have any questions regarding the information contained in this document or Renesas Electronics products, or if you have any other inquiries.

(Note 1) “Renesas Electronics” as used in this document means Renesas Electronics Corporation and also includes its majority-owned subsidiaries.

(Note 2) “Renesas Electronics product(s)” means any product developed or manufactured by or for Renesas Electronics.

RENESAS

Application Note

Implementing a Software I²C Master with the K-Line Microcontroller

Document No. U17206EE1V1AN00
Date Published September 2005

Printed in Germany

NOTES FOR CMOS DEVICES

① VOLTAGE APPLICATION WAVEFORM AT INPUT PIN

Waveform distortion due to input noise or a reflected wave may cause malfunction. If the input of the CMOS device stays in the area between V_{IL} (MAX) and V_{IH} (MIN) due to noise, etc., the device may malfunction. Take care to prevent chattering noise from entering the device when the input level is fixed, and also in the transition period when the input level passes through the area between V_{IL} (MAX) and V_{IH} (MIN).

② HANDLING OF UNUSED INPUT PINS

Unconnected CMOS device inputs can be cause of malfunction. If an input pin is unconnected, it is possible that an internal input level may be generated due to noise, etc., causing malfunction. CMOS devices behave differently than Bipolar or NMOS devices. Input levels of CMOS devices must be fixed high or low by using pull-up or pull-down circuitry. Each unused pin should be connected to V_{DD} or GND via a resistor if there is a possibility that it will be an output pin. All handling related to unused pins must be judged separately for each device and according to related specifications governing the device.

③ PRECAUTION AGAINST ESD

A strong electric field, when exposed to a MOS device, can cause destruction of the gate oxide and ultimately degrade the device operation. Steps must be taken to stop generation of static electricity as much as possible, and quickly dissipate it when it has occurred. Environmental control must be adequate. When it is dry, a humidifier should be used. It is recommended to avoid using insulators that easily build up static electricity. Semiconductor devices must be stored and transported in an anti-static container, static shielding bag or conductive material. All test and measurement tools including work benches and floors should be grounded. The operator should be grounded using a wrist strap. Semiconductor devices must not be touched with bare hands. Similar precautions need to be taken for PW boards with mounted semiconductor devices.

④ STATUS BEFORE INITIALIZATION

Power-on does not necessarily define the initial status of a MOS device. Immediately after the power source is turned ON, devices with reset functions have not yet been initialized. Hence, power-on does not guarantee output pin levels, I/O settings or contents of registers. A device is not initialized until the reset signal is received. A reset operation must be executed immediately after power-on for devices with reset functions.

⑤ POWER ON/OFF SEQUENCE

In the case of a device that uses different power supplies for the internal operation and external interface, as a rule, switch on the external power supply after switching on the internal power supply. When switching the power supply off, as a rule, switch off the external power supply and then the internal power supply. Use of the reverse power on/off sequences may result in the application of an overvoltage to the internal elements of the device, causing malfunction and degradation of internal elements due to the passage of an abnormal current.

The correct power on/off sequence must be judged separately for each device and according to related specifications governing the device.

⑥ INPUT OF SIGNAL DURING POWER OFF STATE

Do not input signals or an I/O pull-up power supply while the device is not powered. The current injection that results from input of such a signal or I/O pull-up power supply may cause malfunction and the abnormal current that passes in the device at this time may cause degradation of internal elements.

Input of signals during the power off state must be judged separately for each device and according to related specifications governing the device.

- **The information in this document is current as of September, 2005. The information is subject to change without notice. For actual design-in, refer to the latest publications of NEC Electronics data sheets or data books, etc., for the most up-to-date specifications of NEC Electronics products. Not all products and/or types are available in every country. Please check with an NEC Electronics sales representative for availability and additional information.**
- No part of this document may be copied or reproduced in any form or by any means without the prior written consent of NEC Electronics. NEC Electronics assumes no responsibility for any errors that may appear in this document.
- NEC Electronics does not assume any liability for infringement of patents, copyrights or other intellectual property rights of third parties by or arising from the use of NEC Electronics products listed in this document or any other liability arising from the use of such products. No license, express, implied or otherwise, is granted under any patents, copyrights or other intellectual property rights of NEC Electronics or others.
- Descriptions of circuits, software and other related information in this document are provided for illustrative purposes in semiconductor product operation and application examples. The incorporation of these circuits, software and information in the design of a customer's equipment shall be done under the full responsibility of the customer. NEC Electronics assumes no responsibility for any losses incurred by customers or third parties arising from the use of these circuits, software and information.
- While NEC Electronics endeavors to enhance the quality, reliability and safety of NEC Electronics products, customers agree and acknowledge that the possibility of defects thereof cannot be eliminated entirely. To minimize risks of damage to property or injury (including death) to persons arising from defects in NEC Electronics products, customers must incorporate sufficient safety measures in their design, such as redundancy, fire-containment and anti-failure features.
- NEC Electronics products are classified into the following three quality grades: "Standard", "Special" and "Specific".
  The "Specific" quality grade applies only to NEC Electronics products developed based on a customer-designated "quality assurance program" for a specific application. The recommended applications of an NEC Electronics product depend on its quality grade, as indicated below. Customers must check the quality grade of each NEC Electronics product before using it in a particular application.
  "Standard": Computers, office equipment, communications equipment, test and measurement equipment, audio and visual equipment, home electronic appliances, machine tools, personal electronic equipment and industrial robots.
  "Special": Transportation equipment (automobiles, trains, ships, etc.), traffic control systems, anti-disaster systems, anti-crime systems, safety equipment and medical equipment (not specifically designed for life support).
  "Specific": Aircraft, aerospace equipment, submersible repeaters, nuclear reactor control systems, life support systems and medical equipment for life support, etc.

The quality grade of NEC Electronics products is "Standard" unless otherwise expressly specified in NEC Electronics data sheets or data books, etc. If customers wish to use NEC Electronics products in applications not intended by NEC Electronics, they must contact an NEC Electronics sales representative in advance to determine NEC Electronics' willingness to support a given application.

(Note)
(1) "NEC Electronics" as used in this statement means NEC Electronics Corporation and also includes its majority-owned subsidiaries.
(2) "NEC Electronics products" means any product developed or manufactured by or for NEC Electronics (as defined above).

M8E 02. 11-1

Regional Information

Some information contained in this document may vary from country to country. Before using any NEC product in your application, please contact the NEC office in your country to obtain a list of authorized representatives and distributors. They will verify:

- Device availability
- Ordering information
- Product release schedule
- Availability of related technical literature
- Development environment specifications (for example, specifications for third-party tools and components, host computers, power plugs, AC supply voltages, and so forth)
- Network requirements

In addition, trademarks, registered trademarks, export restrictions, and other legal issues may also vary from country to country.

NEC Electronics America Inc.
Santa Clara, California
Tel: 408-588-6000
800-366-9782
Fax: 408-588-6130
800-729-9288

NEC Electronics (Europe) GmbH
Duesseldorf, Germany
Tel: 0211-65 03 1101
Fax: 0211-65 03 1327

Sucursal en España
Madrid, Spain
Tel: 091- 504 27 87
Fax: 091- 504 28 60

Succursale Française
Vélizy-Villacoublay, France
Tel: 01-30-67 58 00
Fax: 01-30-67 58 99

Filiale Italiana
Milano, Italy
Tel: 02-66 75 41
Fax: 02-66 75 42 99

Branch The Netherlands
Eindhoven, The Netherlands
Tel: 040-244 58 45
Fax: 040-244 45 80

Branch Sweden
Taeby, Sweden
Tel: 08-63 80 820
Fax: 08-63 80 388

United Kingdom Branch
Milton Keynes, UK
Tel: 01908-691-133
Fax: 01908-670-290

NEC Electronics Hong Kong Ltd.
Hong Kong
Tel: 2886-9318
Fax: 2886-9022/9044

NEC Electronics Hong Kong Ltd.
Seoul Branch
Seoul, Korea
Tel: 02-528-0303
Fax: 02-528-4411

NEC Electronics Singapore Pte. Ltd.
Singapore
Tel: 65-6253-8311
Fax: 65-6250-3583

NEC Electronics Taiwan Ltd.
Taipei, Taiwan
Tel: 02-2719-2377
Fax: 02-2719-5951

Table of Contents

List of Figures

Chapter 1 Introduction

The I²C bus (I²C = IIC = Inter-Integrated Circuit) is a bi-directional two wire clock synchronous bus operating in a master / slave relationship. It consists of a data line (SDA) and a clock line (SCL). The master device always generates the clock. Maximum throughput is 100 Kbit/s for standard devices, 400 Kbit/s for fast mode and in 1998 version 2.0 was introduced, operating at up to 3.4 Mbit/s.

Some NEC microcontrollers are available with I²C hardware; for those parts without an I²C port a collection of software routines are presented here that can be used to create an I²C master with any NEC MCU. Only two bi-directional port pins are needed.

I²C is a registered trademark of Philips Corporation.

Chapter 2 Hardware Arrangement

The I²C bus operates on a wired-AND principle, allowing cascading of any number of devices on a single bus. (In practice the number of devices is limited by the number of device addresses available).
Figure 2-1 shows internal I²C bus interface circuitry in a simplified form:

Figure 2-1: Simplified I²C Driver Architecture



It is clear from the above that some form of pull-ups are required for the output open drain transistor to function correctly.
Figure 2-2 shows how I²C devices are typically interconnected, using a pull-up resistor for SCL (Serial Clock) and SDA (Serial Data). The exact value of these resistors depends on supply voltage, bus capacitance and the number of devices on the bus. The maximum bus capacitance permitted is 400 pF. Active pull-ups can be used in difficult conditions (e.g. where long PCB tracks give rise to high capacitance). The value of 4K7 shown below works satisfactorily for most small systems.
For further information, refer to the Philips publication “The I²C Bus Specification Version 2.1”, January 2000.
The I/O port pins of the NEC MCU may be switched from input to output as required to emulate the arrangement above.

Figure 2-2: I²C Bus Typical Interconnection



Chapter 3 I²C Bus Data Format

3.1 START and STOP Conditions

All data transfers are initiated and terminated with a unique bus condition. A HIGH to LOW transition on the SDA line while SCL is HIGH is considered a START (S) condition while a LOW to HIGH transition on SDA with SCL HIGH is a STOP (P) condition. See Figure 3-1 below.

Figure 3-1: I²C START and STOP Conditions



3.2 Acknowledgement

All I²C byte transfers must end with an acknowledgement (ACK) from the receiving device. This is done by the master releasing the SDA line (i.e. switching it to an input) and transmitting a *ninth* clock pulse. During this ninth clock period the receiving slave device must keep SDA pulled to a stable LOW (ACK). If the receiver is unable to service the transfer it may leave SDA HIGH, this is called a not-acknowledge (NACK) condition. The transmitter can then act upon the NACK, either ending the transfer with a STOP condition or attempting the action again with a repeated START. (Repeated START is the term given to a START condition that appears in the middle of a transfer. It is also used during an I²C read operation, see later.)

Figure 3-2: I²C ACK and NACK Conditions



3.3 Clock Stretching

Sometimes the master device will need to access a slave device that cannot respond immediately to the read or write request. This may be because the slave is busy or it is just an inherently slower device. A clock stretching mechanism is available for this situation: a slave is permitted to hold the SCL line low while it is busy, and then release it so the master can continue the transmission. For example, in Figure 3-3, the first byte could be the address of a byte to be read from a slave device. This is acknowledged by the slave, which may then take a relatively long time to retrieve the data at this address, so it holds SCL low while it does this. The master must poll the SCL line to detect its release; then clocking of SCL may continue.

Figure 3-3: Clock Stretching



The firmware presented later does not allow for clock stretching as it is, but may easily be modified to do so if the application requires it. It is important to note the macros to control SCL will need to be modified so that rather than switching SCL high or low they will switch it from input to output (with the port value always 0) to avoid contention with a slave that is trying to hold the line low, and also to facilitate polling of the SCL line. A timeout timer may also be added to ensure the master does not wait indefinitely should there be a fault in the slave device.

3.4 Data Transfer Format

Data is always transmitted in 8 bit bytes, MSB (Most Significant Byte) first. The first byte to be sent has the slave address in the seven MSB's, followed by the read / write bit, which is set to read and clear to write. The slave address is defined by the device manufacturer and is unique to a particular device, thus allowing many devices of different manufacturer to co-exist on the same bus. Some parts, especially memories, have an address *range* specified at manufacture but leave several pins free for the user to connect to define the exact address within the range, so for example four 2K EEPROMS may be connected to the bus with no additional hardware; their base address defined at manufacture and their individual addresses defined by the logic levels on their address pins. Ten bit addressing is possible with I²C but will not be covered by this application note.

Figure 3-4 shows the data format for a master reading and writing a slave device, and for a combination transfer.

When the master wants to write to the slave, the following happens:

1. Master sends START condition.
2. Master sends slave address with R/W bit CLEAR.
3. Slave issues ACK on ninth clock pulse.
4. Master sends first data byte.
5. Slave issues ACK.
6. Steps 4 and 5 are repeated for all data bytes.
7. Transfer ends with either a STOP condition after the last data byte / ACK pair if the master has no more data to send, or if the slave does not wish to take more data it can inform the master by issuing a NACK after the last data byte. The master then issues a STOP condition as usual to terminate the transfer.

When the master wants to read from the slave, the following happens:

1. Master sends START condition.
2. Master sends slave address with R/W bit SET.
3. Slave issues ACK on ninth clock pulse.
4. Master reads first data byte.
5. Master issues ACK.
6. Steps 4 and 5 are repeated for all data bytes except the last.
7. After reading the last byte, the master issues a NACK to inform the slave there is no more data to be transferred.
8. The master issues a STOP condition.

For the combined transfer (example shown here is write to slave followed by read from slave) the following happens:

1. The slave address and bytes to be *written* are sent in the same manner as for a straightforward write as described above.
2. A *repeated START* is issued by the master followed by the slave address, this time with the R/W bit SET (read).
3. Data is read from the slave in the usual way.
4. The master issues a NACK to indicate to the slave it no longer wishes to read data.
5. The transfer ends with a STOP condition.

Figure 3-4: Data Transfer Format



3.5 EEPROM Read and Write

The firmware listed in this application note can perform, in addition to basic read byte / write byte operations, reads and writes to a 24CXX EEPROM (Electrically Erasable Read Only Memory). More details of the EEPROM can be found in the relevant data sheet, but the format for writing data follows “master writing to slave” in Figure 3-4 above, i.e:

1. Write device address (R/W bit set to WRITE)
2. Write address *within* device
3. Write data byte

To read data, follow “combined” above, i.e:

1. Write device address (R/W bit set to WRITE)
2. Write address *within* device
3. Issue repeated START
4. Write device address (R/W bit set to READ)
5. Read data byte

Chapter 4 Firmware – Flow Diagrams and Description

Before the firmware listing is presented, flow diagrams for each function are shown here.

4.1 START Condition

Figure 4-1: START Condition



4.2 STOP Condition

Figure 4-2: STOP Condition



4.3 Bus Check

Figure 4-3: Bus Check



4.4 Send Byte

Figure 4-4: Send Byte



4.5 Receive Byte

Figure 4-5: Receive Byte



4.6 Get / Put I²C Byte

Figure 4-6: Get / Put I²C Byte



4.7 Send EEPROM Page

Figure 4-7: Send EEPROM Page



4.8 EEPROM Read and Write

Figure 4-8: EEPROM Read and Write



Chapter 5 Firmware – Program Listings

5.1 Main (test) Program

The program was developed with IAR's Embedded Workbench, with i2c_eeprom.c and i2c.msa added to the project under "Options → Files...".
The main program (in C) simply uses the _SEND_EEPROM and _RECV_EEPROM assembly language routines to repeatedly write numbers 0 – 19 to EEPROM locations 0 – 19, and read them back into a buffer as a means of testing and demonstrating the I²C operation. Other subroutines can be called from the users application following the same convention.

```
/*============================================================================
** PROJECT      = I2C_1.prj
** MODULE       = i2c_eeprom.c
** VERSION      = 0.1
** DATE         = 18.03.2001
** LAST CHANGE  = 01.06.2004
** ============================================================================
** Description: Operation as 16-bit timer interrupt
**
** ============================================================================
** Environment:  Device:      uPD78911x
**                            Assembler:            A78000 Version 3.34.2.4
**                            C-Compiler:           ICC78000 Version 3.34.2.4
**                            Linker:               XLINK Version 4.55.9.0
**
** ============================================================================
**      By:            NEC Electronics (Europe) GmbH
**                     Arcadia Strasse 10
**                     D-40472 Duesseldorf
**      and:
**                     NEC Electronics (Europe) GmbH
**                     Cygnus House
**                     Sunrise Parkway
**                     Milton Keynes MK14 6NP
**
** ============================================================================
Changes:
** ============================================================================
*/

/* ============================================================================
** pragma
** ============================================================================
*/
#pragma language = extended

/* ============================================================================
** include
** ============================================================================
*/
#include <in78000.h>
#include "df9116a.h"
#include "i2c.h"

/* ============================================================================
** type definitions (function prototypes)
** ============================================================================
*/
```

```
/* ============================================================================
** variable definitions
** ============================================================================
*/

saddr char count1 = 0, received_data[20];


/* ============================================================================
** variable init
** ============================================================================
*/

void hdwinit (void){

       // port setting
       PM0  = 0xF0;                                  // port 0 = output
       PM1  = 0xFC;                                  // port 1 = output
       PM2  = 0xC0;                                  // port 2 = output
       PM5  = 0xF0;                                  // port 5 = output
       PU0  = 0x00;                                  // no pull up-resistors
       PUB2 = 0x00;

       // clock generator setting
       PCC = 0x00;                           // with speed

}



/* ============================================================================
** main function
** ============================================================================
*/

void main(void){

       hdwinit ();                   // peripheral settings
       _Reset_Bus();


       for(;;){                      // endless loop - main loop

              for (count1 = 0; count1 < 20; count1++){
                     _Send_Eeprom (0, count1, count1);

              }

              for (count1 = 0; count1 < 20; count1++){
                     received_data [count1] = _Recv_Eeprom(0, count1);

              }

       }

}
```

5.2 Header File – i2c.h

```
/*=============================================================================
** PROJECT      = I2C_1.prj
** MODULE       = i2c.h
** VERSION      = 0.1
** DATE         = 20.12.2001
** LAST CHANGE  = 01.06.2004
** ============================================================================
** Description: Header file for the I2C communication
**              needs also i2c.msa file
** ============================================================================
** Environment:  Device:      uPD789xxx
**                            Assembler:            A78000 Version 3.34.2.4
**                            C-Compiler:           ICC78000 Version 3.34.2.4
**                            Linker:               XLINK Version 4.55.9.0
**
** ============================================================================
**      By:            NEC Electronics (Europe) GmbH
**                     Arcadia Strasse 10
**                     D-40472 Duesseldorf
**      and:
**                     NEC Electronics (Europe) GmbH
**                     Cygnus House
**                     Sunrise Parkway
**                     Milton Keynes MK14 6NP
**
** ============================================================================
Changes:
** ============================================================================
*/


extern void _Reset_Bus(void);

extern void _Put_I2C_Byte (unsigned char a,unsigned char d);
extern char _Get_I2C_Byte (unsigned char a);
extern void _Put_I2C_Reg (unsigned char a,unsigned char r,unsigned char d);
extern unsigned char _Get_I2C_Reg (unsigned char a,unsigned char r);
/* the variable names stand for:
                                    a = device-address
                                    r = register-address (address in device)
                                    d = data byte
*/

extern void _Send_Eeprom(unsigned char page,unsigned char a,unsigned char d);
extern unsigned char _Recv_Eeprom(unsigned char page,unsigned char a);
/* the variable names stand for:
                                    address is fixed in the .msa file to 0xA0
                                    page = device page or chip address
                                    a = memory adress
                                    d = data byte
*/



extern unsigned char eeprom_bitreg;

/*
 this variable is declared in the msa file and used for several flags:

    Bit0:  internally used, read flag for i2c communication
    Bit1:  set to 1, if eeprom error occurs after 5 attempts to access the eeprom
    Bit2:  internally used, read/write flag to distinguish between the eeprom read
           and write operations
    Bit3:  set to 1 if the bus is not free
```

```
    Bit4:  set to 1 if i2c communication error occurs
    Bit5:  internally used, read/write flag to distinguish between the i2c read
           and write operations

The error handling has to be done by the C software.
The wait time for the write cycle of the eeprom is done by polling the acknowledge
after sending the device address again (max. 100 times)


*/
```

5.3 Assembly Language Subroutines

```
;NEC Electronics Europe
;General purpose I2C driver routines
;with EEPROM routines


#include <df9116a.h>

 public  _Put_I2C_Reg
 public  _Get_I2C_Reg
 public  _Put_I2C_Byte
 public  _Get_I2C_Byte
 public  _Send_Eeprom
 public  _Recv_Eeprom
 public  _Reset_Bus
 public  eeprom_bitreg

;
;I2C I/O...
;

SDA             equ     P5.1
SCL             equ     P5.0

#define SDAIN           SET1 PM5.1
#define SDAOUT          CLR1 PM5.1
#define SCLIN           SET1 PM5.0
#define SCLOUT          CLR1 PM5.0

;
;flags
;

rw_flag         equ eeprom_bitreg.0
eeprom_error    equ eeprom_bitreg.1
eeprom_read     equ eeprom_bitreg.2
bus_error       equ eeprom_bitreg.3
i2c_error       equ eeprom_bitreg.4
i2c_read        equ eeprom_bitreg.5


wait            equ     100          ;check x times for acknowledge after write


; ============================================================================================

;Macros...

;Set SCL
Set_SCL MACRO
    set1            SCL              ;3 times to guarantee pulse width
    set1            SCL
    set1            SCL
  ENDM

;Clear SCL
Clr_SCL MACRO
    clr1            SCL
    nop
    ENDM
;
;Pulse SCL
Emit_Clock MACRO
    Set_SCL
    Clr_SCL
```

```
    ENDM

; ================================================================================================

;
;variable definition
;
  RSEG UDATA2
  SADDR

bitcount DS 1
eeprom_bitreg ds 1



; ================================================================================================

;
;Start of executable code
;
  RSEG CODE
;
;Subroutines...
;
; ================================================================================================

;Start Sequence
Start:
    SDAOUT
    SCLOUT
    set1    SDA
    Set_SCL
    clr1    SDA
    Clr_SCL
    ret


; ================================================================================================

;Stop Sequence
Stop:
    SDAOUT
    SCLOUT
    clr1    SDA
    Set_SCL
    set1    SDA
    ret

; ================================================================================================

;
;Bus check routine, checks if I2C bus is free
;if not flag bus_error is set
;
Bus_check:
    clr1   bus_error
    clr1   cy
    SCLIN
    SDAIN
    bf     SCL,bus_fault
    bf     SDA,bus_fault    ;jump if bus fault
    SDAOUT
    SCLOUT
    ret
bus_fault:                  ;bus fault
    set1    bus_error       ;set error code
    set1    cy
```

```
    ret

; ================================================================================================

;
;Transmit a byte over the I2C bus
;input: acc contains byte to transmit
;output: cy = 0 if sequence completes
;        cy = 1 if unable to transmit
;on error the i2c error flag is set
;
Xmit_Byte:
       mov     bitcount,#8           ;8 bits to send
xb1:   rolc     a,1
       bnc     xb1a
       set1     SDA                  ;put bit on pin
       br      xb1b
xb1a:  clr1    SDA
xb1b:  Emit_Clock                    ;emit clock pulse
       dbnz    bitcount,xb1          ;loop until done

;setup to accept ACK from slave device
       set1    SDA                    ;release data pin
       Set_SCL                        ;SCL high
       SDAIN
       nop
       bf      SDA,xb2                ;jump if ACK seen
       SDAOUT
       Clr_SCL                        ;drop SCL
       set1    cy                     ;set error code
       set1    i2c_error
       ret
xb2:   SDAOUT
       Clr_SCL                        ;drop SCL
       clr1     cy                    ;set completion code
       clr1     i2c_error
       ret

; ================================================================================================

;
;Receive a byte over the I2C bus
;output: acc contains received byte
;        cy is dummied up with a 0
;
Rec_Byte: mov bitcount,#8             ;8 bits to receive
    SDAIN
zb1:nop
    Set_SCL                           ;SCL high
    nop
    bf     SDA,zb10
    set1     cy               ;pick bit off of pin
    br      zb11
zb10:  clr1  cy
zb11:  rolc     a,1
    Clr_SCL                           ;SCL low
    dbnz    bitcount,zb1      ;more bits to receive?
    clr1    cy                        ;must complete ok
    ret

; ================================================================================================

;
;Public routines...
;

;
```

```
;Reset Bus routine, tries to clear the bus after hang-up
;if no clearance is possible, flag bus error is set
;
_Reset_Bus:
    mov eeprom_bitreg,#0
    mov bitcount,#9
    SDAIN
_reset_loop:bt SDA,Reset_end
    Clr_SCL
    nop
    nop
    Set_SCL
    nop
    dbnz bitcount,_reset_loop
    set1 bus_error
Reset_end: ret

; ============================================================================================

;
;Transmit and Receive routine for adressable data
;Transmit device-address and register-adress over I2C bus
;transmits or receives databyte
;input: r1 contains slave address and contains received data
;       r3 contains register address
;       r2 contains data byte, if transmit is used
;output: cy = 0 if sequence completes
;        cy = 1 if unable to transmit
;

_Get_I2C_Reg:
    set1 i2c_read
    br   xrd1
_Put_I2C_Reg:
    clr1   i2c_read
xrd1:
    call   Bus_check
    bc     xrd_end
    mov    x,a              ;setup slave address
    call   Start            ;set Start condition
    call   Xmit_Byte        ;send slave address
    bc     xrd2             ;jump on error
    mov    a,r3             ;setup register address
    call   Xmit_Byte        ;send register address
    bc     xrd2             ;jump on error
    bt     i2c_read,xrd1b
    mov    a,r2             ;setup data byte
    call   Xmit_Byte        ;go send
    br     xrd2
xrd1b:
    call   Start            ;set repeated Start
    mov    a,x              ;setup slave address
    set1   a.0              ;indicate read operation
    call   Xmit_Byte        ;send slave address again
    bc     xrd2             ;jump on error
    call   Rec_Byte         ;go receive
                            ;store data byte
                            ;sequence complete, return code is already in cy:
    set1   SDA              ;set SDA idle
    Emit_Clock              ;emit clock pulse

;
;set Stop condition, return code is already in cy
xrd2:
    call   Stop             ;set Stop condition
xrd_end:
    ret
```

```
; ================================================================================================

;
;Transmit and receive routine for standard devices
;Transmit address and receives or transmits a data byte over I2C bus
;input: r1 contains slave address and contains received data
;       r3 contains data byte in transmit-mode
;output: cy = 0 if sequence completes
;        cy = 1 if unable to transmit
;
_Get_I2C_Byte:
    set1    i2c_read
    set1    a.0             ;indicate read operation
    br      xrdb1
_Put_I2C_Byte:
    clr1    i2c_read
xrdb1:
    call    Bus_check
    bc      xrdb1_end
                            ;setup slave address
    call    Start           ;set Start condition
    call    Xmit_Byte       ;send slave address
    bc      xrdb2           ;jump on error
    bt      i2c_read,xrdb1b
    mov     a,r3            ;setup data byte
    call    Xmit_Byte       ;send data byte
    br      xrdb2
xrdb1b:
    call    Rec_Byte        ;go receive
                            ;store data byte

;sequence complete, return code is already in cy
    set1    SDA             ;set SDA idle
    Emit_Clock              ;emit clock pulse

;set Stop condition, return code is already in cy
xrdb2:
    call    Stop            ;set Stop condition
xrdb1_end:
    ret

; ================================================================================================

;
;Transmit and Receive routine for serial I2C-EEprom type 24Cxx
;Transmit device-address and memory-adress over I2C bus
;transmits or receives one databyte
;input: r1 contains slave address and contains received data
;       r3 contains register address
;       r2 contains data byte, if transmit is used
;output: cy = 0 if sequence completes
;        cy = 1 if unable to transmit
;on error the eeprom error flag is set
;


_Recv_Eeprom:
        set1    eeprom_read
        br      eep1
_Send_Eeprom:
        clr1    eeprom_read
eep1:   push    rp2
        call    Bus_check
        bc      eep_end
        mov     x,a             ;save eeprom page
        push    rp1
```

```
        pop     rp2             ;free register R2 R3
        mov     r2,#5
eep_loop:
        set1    eeprom_error
        call    Start
        clr1    rw_flag
        mov     a,x
        call    send_eeprom_page
        bc      eep_err
        clr1    eeprom_error
        mov     a,r5            ;setup eeprom address
        call    Xmit_Byte
        bt      eeprom_read,eep2
        mov     a,r4            ;setup eeprom data
        call    Xmit_Byte
        br      eep_err
eep2:
        call    Start
        set1    rw_flag
        mov     a,x
        call    send_eeprom_page
        call    Rec_Byte

eep_err:
        call    Stop
        bt      eeprom_read,eep_err2
wtack:
        call    Start
        mov     R3,#wait
        mov     a,x
        call    send_eeprom_page
        bnc     eep_err1
        dbnz    R3,wtack
eep_err1:
        call    Stop

eep_err2:
        bf      eeprom_error,eep_end
        dbnz    r2,eep_loop

eep_end:
        pop     rp2
        ret


send_eeprom_page:
        and     a,#00000111b
        rol     a,1
        or      a,#10100000b
        clr1    a.0
        bf      rw_flag,send_eeprom_addr
        set1    a.0
send_eeprom_addr:
        call    Xmit_Byte
        ret

; ===============================================================================================


  end


; ===============================================================================================
; ===============================================================================================
```

Chapter 6 Conclusion

This note has given an outline of I²C theory, and has shown how an I²C master can be implemented using an NEC K-line microcontroller that has no dedicated I²C port. Speed of operation is determined largely by device choice and operating frequency, and often operation far below the nominal 100 KHz is acceptable. Although used here to access an external EEPROM, the routines may be used to interface to any I²C device, such as a Real Time Clock (RTC), analog to digital converter etc.

NEC

Facsimile Message

Although NEC has taken all possible steps to ensure that the documentation supplied to our customers is complete, bug free and up-to-date, we readily accept that errors may occur. Despite all the care and precautions we've taken, you may encounter problems in the documentation. Please complete this form whenever you'd like to report errors or suggest improvements to us.

From:

Name

Company

Tel. FAX

Address

Thank you for your kind support.

North America
NEC Electronics America Inc.
Corporate Communications Dept.
Fax: 1-800-729-9288
1-408-588-6130

Europe
NEC Electronics (Europe) GmbH
Market Communication Dept.
Fax: +49(0)-211-6503-1344

Hong Kong, Philippines, Oceania
NEC Electronics Hong Kong Ltd.
Fax: +852-2886-9022/9044

Korea
NEC Electronics Hong Kong Ltd.
Seoul Branch
Fax: 02-528-4411

Taiwan
NEC Electronics Taiwan Ltd.
Fax: 02-2719-5951

Asian Nations except Philippines
NEC Electronics Singapore Pte. Ltd.
Fax: +65-6250-3583

Japan
NEC Semiconductor Technical Hotline
Fax: +81- 44-435-9608

I would like to report the following error/make the following suggestion:

Document title: _____

Document number: _____ Page number: _____

If possible, please fax the referenced page or drawing.

Document Rating	Excellent	Good	Acceptable	Poor
Clarity	☐	☐	☐	☐
Technical Accuracy	☐	☐	☐	☐
Organization	☐	☐	☐	☐

[MEMO]